
 Longitudinal Data Analysis Concepts


The objectives of longitudinal data analysis are to examine and compare responses over time. 
The defining feature of a longitudinal data model is its ability to study changes over time within subjects 
and changes over time between groups. For example, longitudinal models can estimate individual-level (subject-specific) regression parameters and population-level regression parameters.
Longitudinal data sets differ from time series data sets because longitudinal data usually consist of a large number of a short series of time points. However, time series data sets usually consist of a single, long series of time points (Diggle, Heagerty, Liang, and Zeger 2002). For example, the monthly average 
of the Dow Jones Industrials Index for several years is a time series data set, and the efficacy of a drug treatment over time for several patients is a longitudinal data set.

To illustrate the ability of longitudinal models to study changes over time, consider cross-sectional studies in which a single outcome is measured for each subject. In the slide above where each point represents one subject, blood pressure appears to be positively related to age. However, you can reach no conclusions regarding blood pressure changes over time within subjects.

Now expand the cross-sectional study of baseline data to a longitudinal study with repeated measurements over time. The baseline data still show a positive relationship between blood pressure and age. However, now you can distinguish changes over time within subjects from differences among subjects at their baseline or initial starting values. Cross-sectional models cannot make this distinction (Diggle, Heagerty, Liang, and Zeger 2002).

An example of a longitudinal study is the Baltimore Longitudinal Study of Aging (Shock et al. 1984). This is a multidisciplinary observational study where participants return approximately every two years for three days of biomedical and psychological examinations. One objective of the study is to look for markers, which can detect prostate cancer at an early stage. One marker with this potential is the prostate-specific antigen (PSA), which is an enzyme produced by both normal and cancerous prostate cells. 
Its level is related to the volume of prostate tissue. However, an elevated PSA level is not necessarily an indicator of prostate cancer because patients with benign prostatic hyperplasia can also have increased PSA levels. Therefore, researchers have hypothesized that the rate of change in the PSA level might 
be a more accurate method of detecting prostate cancer in the early stages of the disease. A longitudinal model can address this hypothesis.
Another example of a longitudinal study is the Indonesian children’s health study (Sommer 1982). 
In this study more than 3000 children had quarterly medical exams for up to six visits to assess whether they suffered from respiratory or diarrheal infection and xerophthalmia. One objective of the study was 
to determine whether children who had a vitamin A deficiency were at increased risk of respiratory infection.

Special methods of statistical analysis are needed for longitudinal data because the set of measurements on one subject tends to be correlated, measurements on the same subject close in time tend to be more highly correlated than measurements far apart in time, and the variances of longitudinal data often change with time. These potential patterns of correlation and variation might combine to produce a complicated covariance structure. This covariance structure must be taken into account to draw valid statistical inferences. Therefore, standard regression and ANOVA models might produce invalid results because two of the parametric assumptions (independent observations and equal variances) might not be valid.

To illustrate the differences between longitudinal data models and other types of models, consider 
the variance-covariance matrix of the response variable for cross-sectional data. If you have four observations, you would have a 4-by-4 variance-covariance matrix with the variances on the main diagonal and the covariances (a measure of how two observations vary together) on the off-diagonals.
In linear regression with continuous responses, the assumptions are that the responses have equal variances and are independent. Therefore, the variances along the diagonal are equal and the covariances along the off-diagonals are 0.

With longitudinal data, there are now multiple measurements taken on each subject. You not only can examine the differences between subjects, but you can also examine the change within subjects across time. There are still only four subjects and the response is continuous. How does this change your variance-covariance matrix?

For longitudinal data models fit in the MIXED procedure, the number of observations is not the 
number of subjects but rather the number of measurements taken on all the subjects. Because there are three repeated measurements on each subject, you now have 12 observations and a 12-by-12 variance-covariance matrix. For a simple longitudinal model, the matrix is now a block-diagonal matrix in which the observations within each block (the block corresponds to a subject) are assumed to be correlated and the observations outside of the blocks are assumed to be independent. In other words, the subjects are still assumed to be independent of each other and the measurements within each subject are assumed to 
be correlated.

If the observations are positively correlated, which often occurs with longitudinal data, then the variances of the time-independent predictor variables (variables that estimate the group effect or between-subject effect such as gender, race, treatment, and so on) are underestimated if the data are analyzed 
as if the observations are independent. In other words, the Type I error rate (rejecting the null hypothesis
when it is true, also known as a false positive) is inflated for these variables (Dunlop 1994).
Details
Dunlop (1994) shows that the variance of the time-independent predictor variable is


 whereis the correlation between the errors within the subject. If the observations 
are positively correlated within subject, then the variance of the time-independent predictor variable will be underestimated if the data are analyzed as if all observations are independent. 

For time-dependent predictor variables (variables that measure the time effect or within-subject effect such as how the measurements change over time), ignoring positive correlation leads to a variance estimate that is too large. In other words, the Type II error rate (failing to reject the null hypothesis when it is false, also known as a false negative) is inflated for these variables (Dunlop 1994). Because 
the variances of the group effects will be underestimated and the variance of the time effects will 
be overestimated if positive correlation is ignored, it is evident that correlated outcomes must 
be addressed to obtain valid analyses. 
Details
Dunlop (1994) shows that the variance of the time-dependent predictor variable is

 and in this situation, ignoring the positive correlation will lead to a variance estimate that 
is too large.

The linear mixed model allows a very flexible approach to modeling longitudinal data. The data structure is the number of observations equals the number of measurements for all the subjects. This means that 
the data do not have to be balanced. 
An advantage of fitting linear mixed models is that PROC MIXED uses all the complete time measurements in the analysis. This method differs from complete case analysis in which any observation with a missing value across any of the time measurements is dropped from the analysis. The method PROC MIXED uses, called a likelihood-based ignorable analysis, leads to a valid analysis when the missing data are MAR (missing at random, which is a less restrictive assumption than missing completely at random (MCAR)). If the probability of missing for a variable X is related to the values of X itself, even after controlling for the other variables, then the value is not missing at random (NMAR). In other words, the probability of missing depends on the unobserved values. The ignorable analysis is not valid and more complex modeling is required (Verbeke and Molenberghs 1997).
PROC MIXED offers a wide variety of covariance structures. This enables the user to directly address 
the within-subject correlation structure and incorporate it into a statistical model. By selecting a parsimonious covariance model that adequately accounts for within-subject correlations, the user can avoid the problems associated with univariate and multivariate ANOVA using PROC GLM (Littell, Stroup, and Freund 2002).
A value is missing at random (MAR) if the probability that it is missing on a variable X is related to some other measured variable (or variables) in the model but does not depend on any unobserved data after controlling for the observed data. With the MAR assumption, a systematic relationship exists between one or more measured variables and the probability of missing data. MAR is sometimes referred 
to as ignorable missing since the missing data mechanism can be ignored and does not need to be taken into account as part of the modeling process.
A value is missing completely at random (MCAR) if the probability that it is missing is independent 
of the unobserved values. The formal definition of MCAR requires that the probability of missing data 
on a variable X is unrelated to the values of X itself. In other words, the observed data values are a simple random sample of the values that you would have observed if the data had been complete.

The first step in any model-building process is to conduct a thorough exploratory data analysis. 
For longitudinal data this involves plotting the individual measurements over time and fitting a smoothing spline over time. Plotting different groups over time and illustrating cross-sectional and longitudinal relationships are also important steps in exploratory data analysis.
The second step is to fit a complex mean model in PROC MIXED and output the ordinary least squares residuals. These residuals can be used to create a sample variogram, and the pattern in the sample variogram can be helpful in selecting a covariance structure. 
The third step is to fit the linear mixed model in PROC MIXED using the selected covariance structure. Eliminating unnecessary terms and fitting a parsimonious model are important steps in the model building process. After a candidate model is selected, the final steps of the model building process are to evaluate model assumptions and to identify potential outliers.

The first step in any model-building process is exploratory data analysis. In this step you create graphs that expose the patterns relevant to the scientific question. The recommendations on the slide above, given by Diggle, Heagerty, Liang, and Zeger (2002), are used to produce the graphs in this section 
and the section on diagnostics.

A scatter plot of the response versus time is a useful graph. Connecting the repeated measurements 
for each subject over time shows you whether there is a discernible pattern common to most subjects. These individual profiles can also provide some information about between-subject variability. 

For example, the slide above is a graph of weight over time for several subjects. These individual profiles illustrate several important patterns (Diggle, Heagerty, Liang, and Zeger 2002).
1. All of the subjects are gaining weight.
2. The subjects that are the heaviest at the beginning of the study tend to be the heaviest throughout 
the study.
3. The variability of the measurements is smaller at the beginning of the study compared 
to the end of the study.

Besides plotting the response over time, it is also useful to include different subgroups on the same graph to illustrate the relationship between the response and an explanatory variable over time. For example, 
in the slide above it appears that both males and females have decreasing blood pressures over time. However, the slope for the males seems to be more pronounced than the slope for the females, which 
might indicate an interaction between gender and time.

Example:	The human immune deficiency virus (HIV) causes AIDS by attacking an immune cell called the CD4+ cell, which facilitates the body’s ability to fight infection. An uninfected person has approximately 1100 cells per milliliter of blood. Because CD4+ cells decrease in number from the time of infection, a person’s CD4+ cell count can be used to monitor disease progression. A subset of the Multicenter AIDS Cohort Study (Kaslow et al. 1987) was obtained for 369 infected men to examine CD4+ cell counts over time. The data are stored in a SAS data set called long.aids.
These are the variables in the data set:
CD4	CD4+ cell count.
time	time in years since seroconversion (time when HIV becomes detectable).
age	in years relative to arbitrary origin.
cigarettes	packs of cigarettes smoked per day.
drug	recreational drug use (1=yes, 0=no).
partners	number of partners relative to arbitrary origin.
depression	CES-D (a depression scale).
id	subject identification number.
	The data were obtained with permission from Professor Peter Diggle’s website.

The researchers hope to characterize the typical time course of CD4+ cell depletion. This information can clarify the relationship between HIV and the immune system, which might be helpful when counseling infected men.
This observational longitudinal study has unbalanced data because the measurements can occur anytime and the number of measurements can vary across subjects. The linear mixed model using PROC MIXED is the model of choice for this analysis.

Example:	Examine the data in long.aids by producing a line listing report and descriptive statistics for the numeric variables. Also produce graphs of the individual profiles and the group profiles.
/* long01d01.sas */
options nodate;
proc print data=long.aids(obs=17);
   var id cd4 time age cigarettes drug partners depression;   
   title 'Line Listing of CD4+ Data';
run;

                                   Line Listing of CD4+ Data

    Obs      id      CD4      time       age    cigarettes    drug    partners    depression

      1    10002     548    -0.74196    6.57         0          0         5             8
      2    10002     893    -0.24641    6.57         0          1         5             2
      3    10002     657     0.24367    6.57         0          1         5            -1
      4    10005     464    -2.72964    6.95         0          1         5             4
      5    10005     845    -2.25051    6.95         0          1         5            -4
      6    10005     752    -0.22177    6.95         0          1         5            -5
      7    10005     459     0.22177    6.95         0          1         5             2
      8    10005     181     0.77481    6.95         0          1         5            -3
      9    10005     434     1.25667    6.95         0          1         5            -7
     10    10029     846    -1.24025    2.64         0          1         5            18
     11    10029    1102    -0.74196    2.64         0          1         5            18
     12    10029     801    -0.25188    2.64         0          1         5            38
     13    10029     824     0.25188    2.64         0          1         5             7
     14    10029     866     0.76934    2.64         0          1         5            15
     15    10029     704     1.41273    2.64         0          1         5            21
     16    10029     757     1.80698    2.64         0          1         5            25
     17    10029     726     2.42026    2.64         0          1         5            29
The variable age is a time-independent variable and the variables cigarettes, drug, partners, 
and depression are time-dependent variables. The data are unbalanced because the subjects are measured at different time points and the number of measurements is different across subjects.
proc means data=long.aids n min max mean median std;
   var cd4 time age cigarettes drug partners depression;
   title 'Descriptive Statistics for CD4+ Data';
run;

                              Descriptive Statistics for CD4+ Data

                                      The MEANS Procedure

  Variable        N        Minimum        Maximum           Mean         Median        Std Dev
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
  CD4          2376     10.0000000        3184.00    765.1313131    701.5000000    399.3715606
  time         2376     -2.9897330      5.4592740      0.8284246      0.7296370      1.8782282
  age          2376    -11.2900000     29.0800000      2.6359512      1.5100000      7.5039253
  cigarettes   2376              0      4.0000000      0.9890572              0      1.4389639
[bookmark: Demo1_5_pg20]  drug         2376              0      1.0000000      0.7558923      1.0000000      0.4296474
  partners     2376     -5.0000000      5.0000000     -0.0340909     -1.0000000      3.6588315
  depression   2376     -7.0000000     49.0000000      2.4957912              0      9.5863051
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
The outcome variable is CD4 with a range of 10 to 3184. The variable time has a range of nearly 3 years before seroconversion (time when HIV becomes detectable) to 5.5 years after seroconversion. 
The variable age is in years relative to an arbitrary origin. The variable cigarettes measures the number 
of packs smoked per day with a range of 0 (non-smokers) to 4 (heavy smokers). The variable drug 
is a binary variable where 1 means the subject used recreational drugs since the time of the last CD4+ measurement. The mean of 0.76 shows that 76% of the observations had indicated the usage 
of recreational drugs since their last CD4+ cell count measurement. The variable partners measures 
the number of partners relative to an arbitrary origin. The variable depression is a measure of depressive symptoms where a higher score indicates greater depressive symptoms.
To compute descriptive statistics aggregated by subject, an output data set must be created in the MEANS procedure aggregated by subject. A DATA step is used to create a new variable called druguse that indicates whether the subject used recreational drugs at any time during the study period.
proc means data=long.aids noprint nway;
   class id;
   var cd4 age cigarettes drug partners depression;
   output out=subject mean=avgid_cd4 avgid_age 
       avgid_cigarettes avgid_drug avgid_partners avgid_depression;
run;

data subject;
   set subject;
   druguse=(avgid_drug gt 0);
run;

proc means data=subject n min max mean median std;
   var _freq_ avgid_cd4 avgid_age avgid_cigarettes
       avgid_drug druguse avgid_partners avgid_depression;
   title 'Descriptive Statistics for CD4+ Data '
         'Aggregated by Subject';
run;     
Selected PROC MEANS statement option:
NWAY	causes the output data set to have only one observation for each level of the class variable.
                   Descriptive Statistics for CD4+ Data Aggregated by Subject

                                      The MEANS Procedure

  Variable            N       Minimum       Maximum          Mean        Median       Std Dev
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
  _FREQ_            369     1.0000000    12.0000000     6.4390244     6.0000000     2.7141327
  avgid_cd4         369   245.6000000       1979.50   773.7727088   731.4000000   290.4222593
  avgid_age         369   -11.2900000    29.0800000     2.3844444     1.3400000     7.4355005
  avgid_cigarettes  369             0     4.0000000     1.0914632             0     1.3870508
  avgid_drug        369             0     1.0000000     0.7728360     1.0000000     0.3373715
  druguse           369             0     1.0000000     0.9024390     1.0000000     0.2971230
[bookmark: Demo1_5_pg21]  avgid_partners    369    -4.5555556     5.0000000     0.1284397    -0.4000000     2.7544843
  avgid_depression  369    -6.8333333    40.5000000     2.5154360     1.0000000     7.7397510
  ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
The _FREQ_ variable indicates that 369 subjects participated in the study. The average number 
of repeated measures was 6.4 with a range of 1 to 12. The average of the average CD4+ cell count was 773.77 with one subject having the lowest average of 245.6 and another subject having the highest average of 1979.5. The variable druguse indicates that 90% of the participants used recreational drugs during the study period. The average of the average depression score was 2.5 with one subject having 
the lowest average of –6.8 and another subject having the highest average of 40.5.
To create a graph of individual profiles, plot CD4 versus time by subject identification number.
proc sgplot data=long.aids nocycleattrs noautolegend;
   series y=cd4 x=time / group=id 
        lineattrs=(color=blue pattern=1);
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 3500 by 500) label='CD4 Cell Counts';
   title 'Individual Profiles of the CD4+ Data';
run;
Selected PROC SGPLOT statement options:
CYCLEATTRS | NOCYCLEATTRS 	specifies whether plots are drawn with unique attributes in the graph. By default, the SGPLOT procedure automatically assigns unique attributes in many situations, depending on the types of plots that you specify. If the plots do not have unique attributes by default, then the CYCLEATTRS option assigns unique attributes to each plot in the graph. The NOCYCLEATTRS option prevents the procedure from assigning unique attributes. 
NOAUTOLEGEND 	disables automatic legends from being generated. By default, legends are created automatically for some plots, depending 
on their content. This option has no effect if you specify 
a KEYLEGEND statement.
Selected SGPLOT procedure statements:
SERIES	creates a line plot.
XAXIS	specifies options for the X-axis. 
YAXIS	specifies options for the Y-axis. 
Selected SERIES statement options:
X= variable 	specifies the variable for the X-axis. 
Y= variable	specifies the variable for the Y-axis. 
GROUP= variable 	specifies a variable that is used to group the data. A separate 
plot is created for each unique value of the grouping variable. The plot elements for each group value are automatically distinguished by different visual attributes. 
[bookmark: Demo1_5_pg22]LINEATTRS= style-element	specifies the appearance of the series line. You can specify the appearance by using a style element or by using suboptions. 
If you specify a style element, you can in addition specify suboptions to override specific appearance attributes. 
Selected LINEATTRS= options:
COLOR= color 	specifies the color of the line. 
PATTERN= line-pattern 	specifies the line pattern for the line. 
Selected XAXIS statement option:
VALUES= ( value-1 < ...value-n >) 	specifies the values for the ticks on the axis. 

The individual profiles plot is essentially useless. This is a common problem when there are many subjects in a data set. 
A more meaningful plot is an overlay plot of the individual profiles and the average trend. Therefore, 
a smoothed line is fitted using a penalized B-spline curve. The individual profiles serve as a light background and the average trend is a dark line in the foreground. This strategy was suggested by Tufte (1990) where communication of statistical information is enhanced by adding detail in the background.
proc sgplot data=long.aids nocycleattrs noautolegend;
   series y=cd4 x=time / group=id transparency=0.5
[bookmark: Demo1_5_pg23]        lineattrs=(color=cyan pattern=1);
   pbspline y=cd4 x=time / nomarkers smooth=50 nknots=5
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 3500 by 500) label='CD4 Cell Counts'; 
   title 'Individual Profiles with the Average Trend Line';
run;
Selected SGPLOT procedure statement:
PBSPLINE	creates a fitted penalized B-spline curve. 
Selected SERIES statement option:
TRANSPARENCY=	specifies the degree of transparency for the lines and markers. Specify a value from 0.0 (completely opaque) to 1.0 (completely transparent). The default is 0. 
Selected PBSPLINE statement options:
X= numeric-variable 	specifies the variable for the X-axis. 
Y= numeric-variable 	specifies the variable for the Y-axis. 
NOMARKERS	removes the scatter markers from the plot. 
SMOOTH=	specifies a smoothing parameter value. If you do not specify this option, a smoothing value is determined automatically. 
NKNOTS=	specifies the number of evenly spaced internal knots. The default is 100. 
Selected LINEATTRS= option:
THICKNESS=	specifies the thickness of the line. You can also specify the unit of measure. 
The default unit is pixels. 
[bookmark: Demo1_5_pg24]
The average trend shows that the CD4+ cell count is fairly constant around 1000 but drops off around 
the time of seroconversion. The rate of CD4+ cell loss seems to be more rapid immediately after seroconversion. The relationship between CD4+ cell counts and time seems to be cubic in nature.
To highlight the group profiles, smoothed lines representing different subgroups were fitted. The first graph is by recreational drug usage.
proc format;
   value druggrp 0='no recreational drug use'
                 1='recreational drug use';
run;

proc sgplot data=long.aids;
   pbspline y=cd4 x=time / group=drug nomarkers smooth=50 nknots=5 
        lineattrs=(thickness=3) name="drug";
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 1500 by 500) label='CD4 Cell Counts'; 
   keylegend "drug";
   format drug druggrp.;
   title 'Drug Usage Subgroups';
run;
[bookmark: Demo1_5_pg25]Selected SGPLOT procedure statement:
KEYLEGEND	adds a legend to the plot. The argument in the statement (“name”) specifies the names of one or more plots that you want to include in legend. Each name that you specify must correspond to a value that you entered for the NAME= option in a plot statement. If you do not specify a name, then the legend contains references to all of the plots in the graph. 
Selected PBSPLINE statement option:
NAME=	specifies a name for the plot. You can use the name to refer to this plot in other statements. 

There seems to be very little difference in the trends of the two recreational drug groups. 
To define the cigarette usage subgroups, collapse the levels of cigarettes into three groups. Observations with no cigarette usage are in group 1, observations with one to two packs smoked per day are in group 2, and observations with three to four packs smoked per day are in group 3. The new variable ciggroup is added to long.aids for further use.
[bookmark: Demo1_5_pg26]data long.aids;
   set long.aids;
   ciggroup=1*(cigarettes=0)+2*(0<cigarettes<=2)+3*(2<cigarettes<=4);
run;

proc format;
   value cgroup 1='non-smoker'
                2=1 to 2 packs per day'
                3='3 or more packs per day';
run;

proc sgplot data=long.aids;
   pbspline y=cd4 x=time / group=ciggroup nomarkers smooth=50 nknots=5 
        lineattrs=(thickness=3) name="cigarette";
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 1500 by 500) label='CD4 Cell Counts'; 
   keylegend "cigarette";
   format ciggroup cgroup.;
   title 'Cigarette Usage Subgroups';
run;

[bookmark: Demo1_5_pg27]There seems to be a difference between the cigarette usage subgroups. Heavy cigarette users seem to have a much more rapid rate of CD4+ cell loss compared to non-smokers. The difference in the smoothed lines might indicate a time by cigarette interaction.
To define the age subgroups, collapse the levels of age into four groups. Observations in the first quartile are in group 1, observations in the second quartile are in group 2, observations in the third quartile are in group 3, and observations in the fourth quartile are in group 4.
proc rank data=long.aids groups=4 out=ageranks;
   var age;
   ranks agegroup;
run;

proc format;
   value quartile 0='1st quartile'
                  1='2nd quartile'
                  2='3rd quartile'
                  3='4th quartile';
run;

proc sgplot data=ageranks;
   pbspline y=cd4 x=time / group=agegroup nomarkers smooth=50 nknots=5 
        lineattrs=(thickness=3) name="age";
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 1500 by 500) label='CD4 Cell Counts'; 
   keylegend "age";
   format agegroup quartile.;
   title 'Age Subgroups';
run;
Selected PROC RANK statement option:
GROUPS=n	bins the variables into n groups.
Selected RANK procedure statement:
RANKS	names the group indicators in the OUT= data set. If the RANKS statement is omitted, then the group indicators replace the VAR variables in the OUT= data set.
[bookmark: Demo1_5_pg28]
There seems to be no difference between the four age groups with regard to the trend of CD4+ cell loss.
To define the number of partner subgroups, collapse the levels of partners into four groups. 
[bookmark: Demo1_5_pg29]proc rank data=long.aids groups=4 out=partner_ranks;
   var partners;
   ranks partnergroup;
run;

proc sgplot data=partner_ranks;
   pbspline y=cd4 x=time / group=partnergroup nomarkers smooth=50 
                      nknots=5 lineattrs=(thickness=3) name="partner";
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 1500 by 500) label='CD4 Cell Counts'; 
   keylegend "partner";
   format partnergroup quartile.;
   title 'Partner Subgroups';
run;

There seems to be no difference between the four partner groups with regard to the trend of CD4+ cell loss.
[bookmark: Demo1_5_pg30]To define the depression subgroups, collapse the levels of depression into four groups. 
proc rank data=long.aids groups=4 out=depressranks;
   var depression;
   ranks depressgroup;
run;

proc sgplot data=depressranks;
   pbspline y=cd4 x=time / group=depressgroup nomarkers smooth=50 
                      nknots=5 lineattrs=(thickness=3) name="depress";
   xaxis values=(-3 to 5.5 by 0.5) label='Years since Seroconversion';
   yaxis values=(0 to 1500 by 500) label='CD4 Cell Counts'; 
   keylegend "depress";
   format depressgroup quartile.;
   title 'Depression Subgroups';
run;

There seems to be no difference between the four depression groups with regard to the trend of CD4+ cell loss.



One of the recommendations in exploratory data analysis is to display both the cross-sectional and longitudinal relationships between the response variable and the time-dependent explanatory variables. The cross-sectional relationship can be displayed by a scatter plot of the baseline (or initial value) CD4+ cell count versus the baseline explanatory variable values. The longitudinal relationship can be displayed by a scatter plot of the change in CD4+ cell counts (Y at time t – Y at time 1) versus the change in the explanatory variable values (X at time t – X at time 1). Fitting a smooth curve in the scatter plot can indicate whether there is evidence of a relationship (Diggle, Heagerty, Liang, and Zeger 2002).
[bookmark: _Toc314761105][bookmark: _Toc314815010][bookmark: _Toc314815025][bookmark: _Toc332292048][bookmark: _Toc335118803][bookmark: _Toc335638876][bookmark: _Toc336260663][bookmark: _Toc337126791][bookmark: _Toc394910239][bookmark: _Toc395167473][bookmark: _Toc395874963][bookmark: _Toc396295797][bookmark: _Toc424129582][bookmark: _Toc425956074][bookmark: _Toc448301593][bookmark: _Toc473728948][bookmark: _Toc474166309][bookmark: _Toc475961484][bookmark: _Toc477847579][bookmark: _Toc477849840][bookmark: Demo1_6_pg32][bookmark: _Toc217283325][bookmark: _Toc229534613][bookmark: _Toc230424994]Cross-Sectional and Longitudinal Relationships
Example:	Create five cross-sectional scatter plots of the baseline CD4+ cell counts versus the baseline 
of age, recreational drug use, cigarettes, depression score, and number of partners. Also create four longitudinal scatter plots of the change in CD4+ cell counts versus the change in recreational drug use, cigarettes, depression score, and number of partners. Fit a penalized 
B-spline curve in the plots with continuous covariates and a regression line in the plots with binary covariates. 
/* long01d02.sas */
data aids1 aids2;
   set long.aids;
   by id;
   retain basecd4 basedrug basedepress basecig basepart;
   if first.id then
      do;
         basecd4=cd4;
         basedrug=drug;
         basedepress=depression;
         basecig=cigarettes;
         basepart=partners;
         output aids1;
      end;
   if not first.id then 
      do;
        chngcd4=cd4-basecd4;
        chngdrug=drug-basedrug;
        chngdepress=depression-basedepress;
        chngcig=cigarettes-basecig;
        chngpart=partners-basepart;
        output aids2;
      end;
run;
The first step is to create the baseline variables (in aids1) and the difference from time t and baseline variables (in aids2). Because the data are sorted by subject id and time, BY-group processing is used 
to identify the first observation by subject id. The first observation is used to assign the baseline variable values. The RETAIN statement is used to retain the baseline values across the executions of the DATA step. These baseline values are used to create the change in CD4+ cell count and the change in the covariates.
[bookmark: Demo1_6_pg33]proc sgplot data=aids1 noautolegend;
   scatter y=basecd4 x=age / markerattrs=(color=cyan symbol=circle);
   pbspline y=basecd4 x=age / nomarkers smooth=50 nknots=5
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(-12 to 30 by 2) label='Baseline Age';
   yaxis values=(0 to 3000 by 1000) label='Baseline CD4+'; 
   title 'Baseline CD4+ Cells vs. Baseline Age';
run;

There seems to be a slight positive relationship between baseline CD4+ cell counts and the age of 
the patient.
[bookmark: Demo1_6_pg34]proc sgplot data=aids1 noautolegend;
   scatter y=basecd4 x=basedrug / markerattrs=(color=cyan 
                                  symbol=circle);
   reg y=basecd4 x=basedrug / nomarkers 
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(0 to 1 by .1) label='Baseline Recreational Drug Use';
   yaxis values=(0 to 3000 by 1000) label='Baseline CD4+'; 
   title 'Baseline CD4+ Cells vs. Baseline Recreational Drug Use';
run;
Selected SGPLOT procedure statement:
REG	creates a fitted regression line or curve.

There seems to be a slight positive relationship between baseline CD4+ cell count and baseline recreational drug use.
[bookmark: Demo1_6_pg35]proc sgplot data=aids1 noautolegend;
   scatter y=basecd4 x=basedepress / markerattrs=(color=cyan 
                                     symbol=circle);
   pbspline y=basecd4 x=basedepress / nomarkers smooth=50 nknots=5
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(-10 to 50 by 10) label='Baseline Depression Score';
   yaxis values=(0 to 3000 by 1000) label='Baseline CD4+'; 
   title 'Baseline CD4+ Cells vs. Baseline Depression Score';
run;

There seems to a slight upward trend between the baseline values of CD4+ cell counts and the baseline values of the depression scores.
[bookmark: Demo1_6_pg36]proc sgplot data=aids1 noautolegend;
   scatter y=basecd4 x=basecig / markerattrs=(color=cyan 
                                 symbol=circle);
   pbspline y=basecd4 x=basecig / nomarkers smooth=50 nknots=5
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(0 to 4 by 1) label='Baseline Cigarette Usage';
   yaxis values=(0 to 3000 by 1000) label='Baseline CD4+'; 
   title 'Baseline CD4+ Cells vs. Baseline Cigarette Usage';
run;

The graph shows a positive relationship between the baseline values of CD4+ cell counts and the baseline values of number of packs of cigarettes smoked per day. It seems that heavy smokers have higher CD4+ cell counts than non-smokers. This was also shown in the cigarette subgroup profile plot.
[bookmark: Demo1_6_pg37]proc sgplot data=aids1 noautolegend;
   scatter y=basecd4 x=basepart / markerattrs=(color=cyan
                                  symbol=circle);
   pbspline y=basecd4 x=basepart / nomarkers smooth=50 nknots=5
                    lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(-5 to 5 by 1) label='Baseline Partners';
   yaxis values=(0 to 3000 by 1000) label='Baseline CD4+'; 
   title 'Baseline CD4+ Cells vs. Baseline Partners';
run;

There is little evidence of a relationship between baseline CD4+ cell counts and baseline number 
of partners. The uptick in baseline CD4+ at the most extreme left of the plot is due to only one observation.
[bookmark: Demo1_6_pg38]proc sgplot data=aids2 noautolegend;
   scatter y=chngcd4 x=chngdrug / markerattrs=(color=cyan 
                                  symbol=circle);
   pbspline y=chngcd4 x=chngdrug / nomarkers smooth=50 nknots=3
                        lineattrs=(color=blue pattern=1 thickness=3);
   refline 0;
   xaxis values=(-1 to 1 by .1) label='Change in Recreational Drug
                                       Use';
   yaxis values=(-2500 to 2500 by 500) label='Change CD4+'; 
   title 'Change CD4+ Cells vs. Change in Recreational Drug Use';
run;
Selected SGPLOT procedure statement:
REFLINE	creates a horizontal or vertical reference line. 

There seems to be no relationship between the change in CD4+ cell counts and the change in recreational drug use. The only noticeable pattern is that the patients who had no change in their recreational drug use had the smallest decreases in CD4+ cell counts.
[bookmark: Demo1_6_pg39]proc sgplot data=aids2 noautolegend;
   scatter y=chngcd4 x=chngdepress / markerattrs=(color=cyan 
                                     symbol=circle);
   pbspline y=chngcd4 x=chngdepress / nomarkers smooth=50 nknots=5
                       lineattrs=(color=blue pattern=1 thickness=3);
   refline 0;
   xaxis values=(-50 to 50 by 10) label='Change in Depression Score';
   yaxis values=(-2500 to 2500 by 500) label='Change CD4+'; 
   title 'Change CD4+ Cells vs. Change in Depression Score';
run;

There is some evidence that there is a negative relationship between the change in CD4+ cell counts 
and the change in depression score. This implies that a decrease in CD4+ cell counts is associated with 
an increase in depression.
[bookmark: Demo1_6_pg40]proc sgplot data=aids2 noautolegend;
   scatter y=chngcd4 x=chngcig / markerattrs=(color=cyan 
                                 symbol=circle);
   pbspline y=chngcd4 x=chngcig / nomarkers smooth=50 nknots=5
                       lineattrs=(color=blue pattern=1 thickness=3);
   refline 0;
   xaxis values=(-4 to 4 by 1) label='Change in Cigarette Usage';
   yaxis values=(-2500 to 2500 by 500) label='Change CD4+'; 
   title 'Change CD4+ Cells vs. Change in Cigarette Usage';
run;

There is some evidence that there is a positive relationship between the change in CD4+ cell counts 
and the change in the number of packs smoked per day. This implies that a decrease in CD4+ cell counts is associated with a decrease in smoking.
[bookmark: Demo1_6_pg41]proc sgplot data=aids2 noautolegend;
   scatter y=chngcd4 x=chngpart / markerattrs=(color=cyan 
                                  symbol=circle);
   pbspline y=chngcd4 x=chngpart / nomarkers smooth=50 nknots=5
                        lineattrs=(color=blue pattern=1 thickness=3);
   refline 0;
   xaxis values=(-10 to 10 by 2) label='Change in Partners';
   yaxis values=(-2500 to 2500 by 500) label='Change CD4+'; 
   title 'Change CD4+ Cells vs. Change in Partners';
run;

There seems to be a strong positive relationship between the change in CD4+ cell counts and the change in the number of partners. This implies that a decrease in CD4+ cell counts is associated with a decrease in the number of partners.



Careful exploratory data analysis might help you identify scientifically relevant variables to include in your candidate model. In the candidate model for PROC MIXED, the exploratory results indicate to at least include the quadratic and cubic effects of time and the time by cigarette interaction. The results 
of the cross-sectional and longitudinal plots might help you understand the degree of heterogeneity across men in the rate of CD4+ cell count depletion. 

SAS Studio is the new browser-based SAS programming environment that you can use for data exploration and analysis. A tutorial on SAS Studio can be found at: https://support.sas.com/training/tutorial/studio/get-started.html.

Some procedures, such as GLM, are interactive, meaning they remain active until you submit a QUIT statement, or a new PROC or DATA step. You can run these procedures interactively in SAS Studio using the code editor. However, you must enable the interactive mode by using the icon shown above to activate the interactive mode. 

Running SAS Studio in interactive mode starts a new SAS session. This means that library references and macro variables must be defined for each new session. More information can be found at the SAS Studio documentation: http://support.sas.com/software/products/sasstudio/#s1=2


[bookmark: _Toc448301594][bookmark: _Toc473728949][bookmark: _Toc474166310][bookmark: _Toc475961485][bookmark: _Toc477847580][bookmark: _Toc217283326][bookmark: _Toc229534614][bookmark: _Toc230424995][bookmark: _Toc314761106][bookmark: _Toc314815011][bookmark: _Toc314815026][bookmark: _Toc332292049][bookmark: _Toc335118804][bookmark: _Toc335638877][bookmark: _Toc336260664][bookmark: _Toc337126792][bookmark: _Toc394910240][bookmark: _Toc395167474][bookmark: _Toc395874964][bookmark: _Toc396295798][bookmark: _Toc424129583][bookmark: _Toc425956075][bookmark: _Toc477849841][bookmark: Exercise1_7_pg45]Exercises
Conducting an Exploratory Data Analysis
A pharmaceutical firm conducted a clinical trial to examine heart rates among patients. Each patient was subjected to one of three possible drug treatment levels: drug a, drug b, and a placebo. A baseline measurement was taken and the heart rates were recorded at five unequally spaced time intervals: 
1 minute, 5 minutes, 15 minutes, 30 minutes, and 1 hour. The data are stored in the SAS data set long.heartrate.
These are the variables in the data set:
heartrate	heart rate
patient	patient identification number
drug	drug treatment level (a, b, and p)
hours	time point heart rate was recorded (0.01677, 0.08333, 0.25000, 0.5000, 1.000)
baseline	baseline heart rate.
a.    Submit the program long00d01.sas. Print the first 25 observations in the data set long.heartrate. Then create an output data set using PROC MEANS with the mean heart rate for each patient by drug. Also include the baseline heart rate in the data set. Then generate descriptive statistics for the mean patient heart rate and baseline by drug.
1)    Are the patients and measurements of the heart rates sorted?
2)    Do there appear to be any differences in the means of baseline and the mean patient heart rate by drug?
b.    Generate an individual profiles plot with an average trend line using PROC SGPLOT. Use 50 
as the smoothing factor with 5 knots in the PBSPLINE statement.
1)    What is the general pattern of heartrate by hours?

[bookmark: Section_1PT3]

[bookmark: _Toc198629169][bookmark: _Toc217283327][bookmark: _Toc229534615][bookmark: _Toc230424996][bookmark: _Toc314761107][bookmark: _Toc314815012][bookmark: _Toc314815027][bookmark: _Toc332292050][bookmark: _Toc335118805][bookmark: _Toc335638878][bookmark: _Toc336260665][bookmark: _Toc337126793][bookmark: _Toc394910241][bookmark: _Toc395167475][bookmark: _Toc395874965][bookmark: _Toc396295799][bookmark: _Toc424129584][bookmark: _Toc425956076][bookmark: _Toc448301595][bookmark: _Toc473728950][bookmark: _Toc474166311][bookmark: _Toc475961486][bookmark: _Toc477847581][bookmark: _Toc477849842]Chapter Summary
The objectives of longitudinal data analysis are to examine and compare responses over time. 
The defining feature of a longitudinal data model is its ability to study changes over time within subjects and changes over time between groups.
Special methods of statistical analysis are needed for longitudinal data because the set of measurements on one subject tend to be correlated, measurements on the same subject close in time tend to be more highly correlated than measurements far apart in time, and the variances of longitudinal data often change with time. These potential patterns of correlation and variation might combine to produce a complicated covariance structure. This covariance structure must be taken into account to draw valid statistical inferences. Therefore, standard regression and ANOVA models might produce invalid results because two of the parametric assumptions (independent observations, equal variances) might not hold.
If the observations are positively correlated, which often occurs with longitudinal data, then the variance of the time-independent predictor variables are underestimated if the data are analyzed as if the observations are independent. In other words, the Type I error rate is inflated for these variables.
For time-dependent predictor variables, ignoring positive correlation leads to a variance estimate that is too large. In other words, the Type II error rate is inflated for these variables.
The linear mixed model allows a flexible approach to modeling longitudinal data. The linear mixed model
handles unbalanced data with unequally spaced time points and subjects observed at different time points
uses all the available data in the analysis
directly models the covariance structure
provides valid standard errors and efficient statistical tests.
The first step in any model-building process is exploratory data analysis. In this step you create graphs that expose the patterns relevant to the scientific question. General recommendations are
graph as much of the relevant raw data as possible
highlight aggregate patterns of potential scientific interest
identify both cross-sectional and longitudinal patterns
identify unusual individuals or observations.
A meaningful plot is an overlay plot of the individual profiles and the average trend. A smoothed line representing the average trend can be fitted using a spline routine. The individual profiles can serve as a light background and the average trend can be a dark line in the foreground.

Solutions

1. Conducting an Exploratory Data Analysis
A pharmaceutical firm conducted a clinical trial to examine heart rates among patients. Each patient was subjected to one of three possible drug treatment levels: drug a, drug b, and a placebo. A baseline measurement was taken and the heart rates were recorded at five unequally spaced time intervals: 1 minute, 5 minutes, 15 minutes, 30 minutes, and 1 hour. The data are stored in the SAS data set long.heartrate.
These are the variables in the data set:
heartrate	heart rate
patient	patient identification number
drug	drug treatment level (a, b, and p)
hours	time point heart rate was recorded (0.01677, 0.08333, 0.25000, 0.5000, 1.000)
baseline	baseline heart rate.
a.    Submit the program long00d01.sas. Print the first 25 observations in the data set long.heartrate. Then create an output data set using PROC MEANS with the mean heart rate for each patient by drug. Also include the baseline heart rate in the data set. Then generate descriptive statistics for the mean patient heart rate and baseline by drug.
proc print data=long.heartrate(obs=25);
   title 'Line Listing of Heart Rate Data';
run;
   
[bookmark: Solution1_8_pg49]                                Line Listing of Heart Rate Data

                   Obs    patient    drug    baseline     hours     heartrate

                     1      201       p         92       0.01667        76
                     2      201       p         92       0.08333        84
                     3      201       p         92       0.25000        88
                     4      201       p         92       0.50000        96
                     5      201       p         92       1.00000        84
                     6      202       b         54       0.01667        58
                     7      202       b         54       0.08333        60
                     8      202       b         54       0.25000        60
                     9      202       b         54       0.50000        60
                    10      202       b         54       1.00000        64
                    11      203       p         84       0.01667        86
                    12      203       p         84       0.08333        82
                    13      203       p         84       0.25000        84
                    14      203       p         84       0.50000        86
                    15      203       p         84       1.00000        82
                    16      204       a         72       0.01667        72
                    17      204       a         72       0.08333        68
                    18      204       a         72       0.25000        68
                    19      204       a         72       0.50000        78
                    20      204       a         72       1.00000        72
                    21      205       b         80       0.01667        84
                    22      205       b         80       0.08333        84
                    23      205       b         80       0.25000        96
                    24      205       b         80       0.50000        92
                    25      205       b         80       1.00000        72

1)   The patients and the measurement times are sorted.
proc means data=long.heartrate noprint nway;
   id baseline;
   class patient drug;
   var heartrate;
   output out=subject mean=avgpat_heart;
run;

proc means data=subject n min max mean median std;
   class drug;
   var avgpat_heart baseline;
   title 'Descriptive Statistics for Heart Rate Data Aggregated by '
         'Subject';
run;
   
[bookmark: Solution1_8_pg50]             Descriptive Statistics for Heart Rate Data Aggregated by Subject

                                      The MEANS Procedure

              N
 drug       Obs   Variable        N        Minimum        Maximum           Mean         Median
 ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
 a            8   avgpat_heart    8     56.0000000     91.2000000     77.5000000     76.8000000
                  baseline        8     60.0000000    100.0000000     80.7500000     81.0000000

 b            8   avgpat_heart    8     60.4000000     93.6000000     83.6500000     86.8000000
                  baseline        8     54.0000000    104.0000000     83.2500000     88.0000000

 p            8   avgpat_heart    8     66.8000000     88.4000000     78.8500000     81.6000000
                  baseline        8     68.0000000    102.0000000     84.7500000     86.0000000
 ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
                                       N
                          drug       Obs   Variable            Std Dev
                          ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ
                          a            8   avgpat_heart     12.1245913
                                           baseline         12.8257553

                          b            8   avgpat_heart     10.5536723
                                           baseline         14.9642431

                          p            8   avgpat_heart      7.9840913
                                           baseline         11.2090525
                          ƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒƒ

2)   Patients on the placebo have the highest baseline mean while patients on drug b have the highest mean heart rate. The differences are relatively small across treatment groups.
b.    Generate an individual profiles plot with an average trend line using PROC SGPLOT. Use 50 as the smoothing factor with 5 knots in the PBSPLINE statement.
[bookmark: Solution1_8_pg51]proc sgplot data=long.heartrate nocycleattrs noautolegend;
   series y=heartrate x=hours / group=patient lineattrs=(color=cyan 
                                pattern=1);
   pbspline y=heartrate x=hours / nomarkers smooth=50 nknots=5
        lineattrs=(color=blue pattern=1 thickness=3);
   xaxis values=(0 to 1 by 0.1) label='Time Point Heart Rate was 
                                       Recorded in Hours';
   yaxis values=(40 to 120 by 10) label='Heart Rate';
   title 'Individual Profiles of the Heart Rate Data';
run;

1)   The average heart rate appears to decrease over time. 

Cross-Sectional Analysis


image4.emf
Copyright © SAS Institute Inc. All rights reserved.

6

Longitudinal Analysis

Examples of Longitudinal Studies

•

Examine the rate of change in the prostate-specific antigen level to detect 

prostate cancer in the early stages of the disease.

•

Examine children over time to estimate the increase in risk of respiratory 

infection for those who are vitamin A deficient while controlling for other 

factors. 

Problems with OLS Regression

Models that assume the observations are independent might be 

inappropriate for longitudinal data because

•

measurements taken on the same subject tend to be more similar than 

measurements taken on different subjects

•

measurements taken close in time on the same subject tend to be more 

similar than measurements taken far apart in time

•

the variances of longitudinal data often change with time.

Variance-Covariance Matrix 

for OLS Regression

Longitudinal Data

Variance-Covariance Matrix 

for Longitudinal Data

Effect on 

Time-Independent Predictor Variables

Accounting for 

Positive Correlation



ˆ

Ignoring Positive 

Correlation

Effect on 

Time-Dependent Predictor Variables

Ignoring Positive 

Correlation

Statistical Method for Repeated Continuous 

Responses

The linear mixed model fit by PROC MIXED will be used in this course. The 

strengths are as follows:

•

Handles unbalanced data with unequally spaced time points and subjects 

observed at different time points 

•

Uses all the complete time measurements in the analysis

•

Directly models the covariance structure

•

Provides valid standard errors and efficient statistical tests

Model-Building Strategies

•

Conduct an exploratory data analysis by illustrating the cross-sectional and 

longitudinal relationships in the data.

•

Fit a complex mean model and output OLS residuals.

•

Use the OLS residuals to create a sample variogramto help select a 

covariance structure.

•

Eliminate unnecessary terms in the complex mean model.

•

Evaluate model assumptions and identify potential outliers.

1.01 Multiple Choice Poll

For time-independent predictor variables, if you ignore the positive 

correlation among the repeated measurements, which of the following is 

true?

a. The parameter estimates are biased downward.

b. The parameter estimates are biased upward.

c. The standard errors are underestimated.

d. The standard errors are overestimated.

17

Objectives

•

Graph individual and group profiles.

•

Illustrate how to identify cross-sectional and longitudinal patterns.

Recommendations

•

Graph as much of the relevant raw data as possible.

•

Highlight aggregate patterns of potential scientific interest.

•

Identify both cross-sectional and longitudinal patterns.

•

Identify unusual individuals or observations.

Individual Profiles

Plotting observed profiles over time

•

helps identify general trends within subjects

•

might detect nonlinear change over time

•

provides information about the variability at given times.

Individual Profiles

Group Profiles

CD4+ Cell Numbers Data Set

Years since Seroconversion

CD4+ Cell 

Numbers

Objectives of CD4+ Cell Numbers Study

•

Estimate the average time course of CD4+ cell depletion.

•

Estimate the time course for individual men.

•

Characterize the degree of heterogeneity across men in the rate of 

progression.

•

Identify factors that predict CD4+ cell changes.

Cross-Sectional versus Longitudinal Relationship

Scatter plot of cross-sectional relationship

•

baseline Y versus baseline X

Scatter plot of longitudinal relationship

•

change in Y versus change in X

Summary of Exploratory Data Analysis

•

There seems to be a cubic relationship between CD4+ cell count and time.

•

The group profile plots show a time by cigarette usage interaction.

•

The cross-sectional plots show a positive relationship between the baseline 

CD4+ cell counts and the baseline cigarette usage.

•

The longitudinal plots show a positive relationship between the change in 

CD4+ cell counts and the change in the number of partners.

SAS Studio

SAS Studio is the new browser-based SAS programming environment. 

Interactive Mode

Some SAS procedures, such as PROC GLM, are interactive. That means they 

remain active until you submit a QUIT statement, or until you submit a new 

PROC or DATA step.

In SAS Studio, you can use the code editor to run these procedures, as well as 

other SAS procedures, in interactive mode.

By default, SAS Studio does not 

run in interactive mode.

This icon in SAS Studio toggles 

interactive mode on and off.

Considerations for Running in 

Interactive Mode

•

Interactive mode starts a new SAS session.

•

Librefsand macro variables used in the course must  be defined for each 

new SAS session.

SAS Studio Documentation: 

http://support.sas.com/software/products/sasstudio/#s1=2

1.02 Multiple Choice Poll

What is the general pattern of heart rate by hours?

a. The average heart rate appears to be increasing over time.

b. The average heart rate appears to be decreasing over time.

c. The average heart rate appears to have a quadratic relationship with 

time.

d. The average heart rate appears to be constant over time.

36

1.01 Multiple Choice Poll –Correct Answer

For time-independent predictor variables, if you ignore the positive 

correlation among the repeated measurements, which of the following is 

true?

a. The parameter estimates are biased downward.

b. The parameter estimates are biased upward.

c. The standard errors are underestimated.

d. The standard errors are overestimated.

1.02 Multiple Choice Poll –Correct Answer

What is the general pattern of heart rate by hours?

a. The average heart rate appears to be increasing over time.

c. The average heart rate appears to have a quadratic relationship with 

time.

d. The average heart rate appears to be constant over time.

Objectives

•

Understand how longitudinal data differ from cross-sectional data.

•

Appreciate the merits of longitudinal data analysis.

•

Explain the consequences of ignoring correlated observations.

•

Document the strengths of the linear mixed model.

Longitudinal Data Analysis

•

The defining feature is that repeated measurements are taken on a subject 

through time.

•

The models can distinguish changes over time within subjects from 

differences between subjects at their baseline levels and can also study 

changes over time between groups. 

•

The models can estimate individual-level (subject-specific) regression 

parameters and population-level regression parameters.


